NAG Library Routine Document

F04AJF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F04AJF calculates the approximate solution of a set of real linear equations with multiple right-hand sides, $AX = B$, where A has been factorized by F03AFF.

2 Specification

```
SUBROUTINE F04AJF (N, IR, A, LDA, P, B, LDB)

INTEGER            N, IR, LDA, LDB
REAL (KIND=nag_wp) A(LDA,N), P(N), B(LDB,IR)
```

3 Description

To solve a set of real linear equations $AX = B$, F04AJF must be preceded by a call to F03AFF which computes an LU factorization of A with partial pivoting, $PA = LU$, where P is a permutation matrix, L is lower triangular and U is unit upper triangular. The columns x of the solution X are found by forward and backward substitution in $Ly = Pb$ and $Ux = y$, where b is a column of the right-hand sides.

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer–Verlag

5 Parameters

1: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N $\geq$ 1.

2: IR – INTEGER *Input*

On entry: r, the number of right-hand sides.

3: A(LDA,N) – REAL (KIND=nag_wp) array *Input*

On entry: details of the LU factorization, as returned by F03AEF.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F04AJF is called.

Constraint: LDA $\geq$ N.

5: P(N) – REAL (KIND=nag_wp) array *Input*

On entry: details of the row interchanges as returned by F03AFF.

6: B(LDB,IR) – REAL (KIND=nag_wp) array *Input/Output*

On entry: the n by r right-hand side matrix B.

On exit: B is overwritten by the solution matrix X.

7: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F04AJF is called.

Constraint: LDB $\geq$ N.

6 Error Indicators and Warnings

If an error is detected in an input parameter F04AJF will act as if a soft noisy exit has been requested (see Section 3.3.4 in the Essential Introduction).

7 Accuracy

The accuracy of the computed solutions depends on the conditioning of the original matrix. For a detailed error analysis see page 106 of Wilkinson and Reinsch (1971).

8 Further Comments

The time taken by F04AJF is approximately proportional to n^2r.

9 Example

This example solves the set of linear equations $AX = B$ where

$$A = \begin{pmatrix} 33 & 16 & 72 \\ -24 & -10 & -57 \\ -8 & -4 & -17 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} -359 \\ 281 \\ 85 \end{pmatrix}.$$

9.1 Program Text

```
    Program f04ajfe

!     F04AJF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: f03aff, f04ajf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: d1, eps
      Integer                          :: i, id, ifail, ir, lda, ldb, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:), b(:,:), p(:)
!     .. Executable Statements ..
      Write (nout,*) 'F04AJF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      ir = 1
      lda = n
      ldb = n
      Allocate (a(lda,n),b(ldb,ir),p(n))
      Read (nin,*)(a(i,1:n),i=1,n)
```

```
!     ifail: behaviour on error exit
!             =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
!     Crout decomposition
      Call f03aff(n,eps,a,lda,d1,id,p,ifail)

      Read (nin,*)(b(i,1:ir),i=1,n)

!     Approximate solution of linear equations
      Call f04ajf(n,ir,a,lda,p,b,ldb)

      Write (nout,*) ' Solution'
      Do i = 1, n
        Write (nout,99999) b(i,1:ir)
      End Do

99999 Format (1X,8F9.4)
    End Program f04ajfe
```

9.2 Program Data

```
F04AJF Example Program Data
  3                    : n
   33   16   72
  -24  -10  -57
   -8   -4  -17
 -359  281   85        : matrices A and B
```

9.3 Program Results

```
 F04AJF Example Program Results

  Solution
    1.0000
   -2.0000
   -5.0000
```